

Capitolul 7

Interfața ASM - limbaje de nivel înalt

O parte importantă a programării în limbaj de asamblare o constituie legarea modulelor de program ASM cu module de program dezvoltate în limbajde de nivel înalt. În acest mod, se poate crește eficiența aplicațiilor, prin dezvoltarea modulelor critice (care se utilizează foarte des) în ASM.

În acest capitol va fi exemplificată dezvoltarea de aplicații mixte în limbaj de asamblare și în limbajul C. Alegerea limbajului C ca limbaj reprezentativ de nivel înalt este justificată de următoarele proprietăți:

- răspândirea foarte mare a limbajului C, ca limbaj destinat atât programelor de sistem cât și celor aplicative;
- existența unor medii de dezvoltare evolute, total compatibile cu asamblările 8086 (un exemplu semnificativ fiind familia de produse Borland);
- caracterul modular al limbajului C, prin care se permite compilarea separată a modulelor sursă;
- compatibilitatea totală a modulelor obiect (un modul obiect rezultat în urma compilării unui text sursă C are exact același format ca un modul obiect rezultat în urma asamblării unui text ASM).

Se va utiliza deci mediul de dezvoltare Borland (compilatorul bcc, editorul de legături tlink, asamblorul tasm, respectiv mediul integrat bc).

7.1 Transferul parametrilor. Simboli externi

În mod concret, interfața dintre module C și module ASM va consta din apeluri de funcții ASM din C și invers, respectiv din accesul din C la date definite în ASM și invers.

Compilatorul Borland C realizează transferul parametrilor prin stivă, în ordinea de la dreapta la stânga a listei de parametri. Descarcarea stivei este făcută de către modulul apelant. Întoarcerea de tipuri simple de date din funcții se realizează prin registrul acumulator, eventual extins (deci prin AL, AX, sau DX:AX, corespunzător unui tip pe 1, 2 sau 4 octeți). Tipurile reale (float, double, long double) sunt întoarse printr-o zonă specială a bibliotecii de virgulă mobilă sau în vîrful stivei coprocesorului matematic. Pentru a nu complica interfața, vom considera că funcțiile C folosite în ASM nu întorc valori reale. Dacă este absolut necesar, se poate adopta soluția ca funcțiile C să întoarcă pointeri la variabile reale statice.

Un alt caz complicat este transferul structurilor și uniunilor, care este definit în C prin copierea bit cu bit a tuturor membrilor, atât la transfer spre funcție cât și la întoarcere din funcție. Este și aici de preferat soluția mult mai eficientă de a transfera sau întoarce un pointer către structura sau uniunea respectivă.

Numele simbolilor externi în C (funcții și variabile externe) sunt generate implicit cu caracterul _ (subliniere) ca prim caracter. De exemplu, simbolul var este generat ca _var și vizibil ca atare într-un modul ASM. Pentru variabile, numele este sinonim cu adresa unde este memorată variabila.

Deoarece, în C, vizibilitatea externă este permisă numai variabilelor definite la nivel exterior (în afara tuturor funcțiilor), acestea sunt implicit alocate static,

deci au adrese fixe de memorie. Dacă, de exemplu, var este un întreg definit în C, atunci el poate fi citit în ASM printr-o instrucțiune de tipul mov ax, _var.

Trebuie cunoscute dimensiunile fizice ale tipurilor de bază din C. Acestea sunt: char (1 octet), int (2 octeți), short (2 octeți), long (4 octeți), float (4 octeți), double (8 octeți), long double (10 octeți), pointeri (2 sau 4 octeți, funcție de modelul de memorie folosit).

Trebuie ținut seama și de faptul că operațiile cu stiva sunt operații pe 16 biți, deci în cazul în care se transmite un char la o funcție, se pune pe stivă un cuvânt de 16 biți, cu partea mai semnificativă 0. Este recomandabil ca, asemenea funcțiilor din biblioteca C standard, variabilele de tip char să fie transmise și întoarse la și de la funcții ca variabile de tip unsigned int, ceea ce asigură protecție la o eventuală extensie de semn la 16 biți. De exemplu, variabila char '\x81' devine, prin extensie de semn, întregul 0xff81, dar variabila unsigned char '\x81' devine întregul 0x0081. Extensia de semn poate apărea dacă tipul char este implicit signed (ceea ce se întâmplă la Borland C). Există opțiune de compilare care face ca tipul char să fie implicit unsigned, dar este bine să se scrie programe care să nu depindă de această opțiune.

Un alt fapt care trebuie avut în vedere este că C-ul este un limbaj de tip case-sensitive, deci contează dacă identificatorii sunt scriși cu litere mici sau mari. Trebuie forțată la TASM opțiunea /ml sau /mx care generează simbolii ținând cont de acest fapt.

Să considerăm următoarea secvență C:

```
include <stdio.h>
int n = 5;
char sir[] = "Un sir";
void main(void)
{
    printf("%s %d\n", sir, n);
}
```

Presupunând modelul de memorie small (adică toate adresele sunt pe 2 octeți), imaginea stivei la intrarea în funcția printf este cea din Figura 7.1.




Figura 7.1 Imaginea stivei la un apel al funcției printf

Apelul lui printf din ASM se poate face cu secvența ASM de mai jos (echivalentă cu secvența C):

```
.model small
extrn printf: near
.data
    n          dw      5
    sir        db      'Un sir', 0
    format     db      '%s %d\n', 0
.code
main proc near
    push  n
    lea   ax, sir
    push  ax
    lea   ax, format
    push  ax
    call  near ptr printf
    add   sp, 6
    retn
main endp
```

7.2 Dezvoltarea în mediu integrat sau prin linii de comandă

Sunt posibile două moduri de dezvoltare a aplicației: în mediul integrat C sau prin linie de comandă.

În mediul integrat (lansat prin comanda bc) trebuie definit un proiect care să specifică explicit modulele în limbaj de asamblare. Definirea proiectului permite specificarea programului translator (Compiler sau Assembler).

Astfel, se poate lucra în regim de editare, compilare, depanare, etc., cu ambele tipuri de module (ASM și C), mediul integrat recunoscând modulele scrise în ASM. Acest context este foarte util pentru depanare, deoarece toate facilitățile specifice mediilor integrate de dezvoltare pentru limbaje de nivel înalt (rulare pas cu pas, vizualizarea permanentă a unor variabile etc.), se aplică și modulelor scrise în ASM.

O altă variantă de depanare o constituie Turbo Debugger-ul, care poate fi apelat direct sau din mediul integrat, acesta recunoscând, de asemenea, atât module ASM, cât și module C.

Opțiunile necesare la compilarea programelor C (cea mai importantă fiind modelul de memorie), se stabilesc prin comanda **Options** a mediului integrat.

A doua posibilitate este compilarea, respectiv asamblarea separată a modulelor sursă, prin lansări ale compilatorului bcc și ale asamblorului tasm de la consolă și apoi legarea explicită a modulelor obiect, prin lansarea editorului de legături tlink. În acest caz, opțiunea de model de memorie pentru modulele C se stabilește în linia de apel a compilatorului (opțiunea -mx, unde x este un caracter ce identifică modelul). De asemenea trebuie precizată opțiunea -c (compile only), astfel încât compilatorul C să producă numai modul obiect. Tot în linia de comandă se precizează căile pentru fișierele header și pentru biblioteci (opțiunile -I și -L).

Când se lucrează în regim de linie de comandă, trebuie ținut seama de structura unui program C. Pe lângă modulele definite de utilizator, trebuie să adăugăm modulul de initializare c0x (se pune pe prima poziție în lista de module obiect la tlink) și bibliotecile C (aflate în subdirectorul \lib), care se specifică după numele fișierului executabil. Atât modulul de initializare cât și

bibliotecile sunt specifice modelului de memorie folosit, fiind identificate prin ultima literă a numelui fișierului respectiv.

Bibliotecile care trebuie specificate depind de funcțiile utilizate în modulele C. Practic, se va lega totdeauna biblioteca cx.lib și, eventual, mathx.lib, emu.lib (pentru emularea operațiilor în virgulă mobilă), fp87.lib (dacă sistemul are coprocesor), graphics.lib (dacă s-au folosit funcții grafice). În specificarea numelor fișierelor care conțin modulul de initializare și bibliotecile respective, caracterul x depinde de modelul de memorie folosit.

Este utilă o vizualizare a conținutului subdirectorului \lib din implementarea Borland C, pentru a identifica bibliotecile C. Este de asemenea utilă vizualizarea opțiunilor posibile de compilare, asamblare și link-editare, ca și sintaxa acestor comenzi (se poate obține prin bcc, tasm sau tlink urmate imediat de <Enter>).

În cazul în care modulul de program principal este scris în limbaj de asamblare, trebuie ținut seama de faptul că main este o funcție ca oricare alta. De fapt, modulul de program principal este modulul precompilat c0, care, după ce face o serie de initializări, apelează procedura main. Din această cauză, dacă funcția main este scrisă în ASM, ea trebuie declarată ca o procedură și specificată ca simbol public:

```
public _main
    _main proc
        ...
        ret
    _main endp
end
```

Procedura _main trebuie încheiată prin instrucțiunea ret, ca orice procedură, iar modulul de program respectiv nu trebuie să conțină etichetă de start. Încheierea execuției se va face la return-ul din _main, controlul revenind modulului c0, care a apelat _main și care este răspunzător de ieșirea în sistemul de operare. Nu este necesară o ieșire în DOS din procedura _main. De asemenea, nu este necesară nici initializarea registrelor DS și ES, deoarece este făcută în modulul c0. Modulul c0 definește și o stivă minimă (dacă modelul nu este tiny), care se poate mări, dacă este necesar, prin directiva ASM .stack.

În cazul în care fuția cmain este scrisă în C, modulele ASM vor conține practic proceduri și definiții de date, care trebuie declarate publice (cu ajutorul directivei ASM public). De asemenea, dacă modulele ASM apelează funcții C, acestea trebuie declarate în ASM ca simboli externi (cu ajutorul directivei ASM extrn).

7.3 Modele de memorie

Modelele de memorie corespund organizării segmentelor de date, cod și stivă, specifice adresării de la procesoarele INTEL. Atât asamblorul TASM cât și compilatorul Borland C recunosc aceleași modele de memorie, generând segmente cu același nume. La TASM, acest lucru corespunde folosirii directivelor de definire simplificată a segmentelor. Această compatibilitate totală simplifică foarte mult efortul de dezvoltare al aplicațiilor mixte.

În ASM, modelul se specifică prin directiva .model, iar în C prin opțiunea Options/Compiler/Code generation.../Model în mediul integrat sau prin opțiunea -mx în linia de comandă, la lansarea compilatorului.

Modelele disponibile sunt tiny, small, compact, medium, large și huge. Funcție de aceste modele, se vor genera apeluri de funcții implicit de tip far sau near și pointeri la funcții sau la date implicit de tip far sau near.

Implementările Borland C permit definirea explicită a pointerilor și funcțiilor de tip far sau near, dar acest lucru nu este recomandabil. Este indicat să fie lăsat compilatorul să genereze implicit pointeri și apeluri de funcții, conform modelului de memorie definit. Dacă modulele C includ fișierele header predefinite, în care se găsesc prototipurile pentru funcțiile de bibliotecă folosite, apelurile de funcții și legarea cu modulele de bibliotecă se vor realiza în mod corect.

În modulele ASM se va folosi același model de memorie, dar trebuie ținut cont de tipurile pointerilor și funcțiilor definite

sau folosite în modulele C, deoarece, în ASM, accesul la date, funcții, adrese de segment, offset-uri, etc. se face explicit. Se recomandă ca, în ASM, să se folosească explicit apeluri de tip near/far, definiții de proceduri explicate near/far și return explicit de tip near/far, deși asamblorul generează corect apeluri și definiții de proceduri implicite, pe baza modelului ales. Totuși, pentru a scoate în evidență tipurile respective, este bine să se folosească variante explicate în ASM.

Nu se recomandă în nici un caz ca, în aceeași aplicație, module diferite de program să aibă modele diferite de memorie.

Portabilitatea modulelor C la schimbarea modelului de memorie nu necesită practic nici un efort de întreținere. Portabilitatea modulelor ASM față de modelele de memorie este însă o problemă dificilă. Apelurile de proceduri și instrucțiunile de revenire pot fi gestionate corect de către asamblor. Restul operațiilor trebuie gestionate explicit, eventual prin directive de asamblare condiționată.

Să presupunem că transmitem unei proceduri un parametru de tip adresă de variabilă (de date) și dorim să definim o structură de acces la stivă, care să nu depindă de modelul de memorie utilizat. Soluția constă în folosirea directivelor de asamblare condiționată și a simbolului predefinit @MODEL (vezi 4.2).

```
_TINY_      equ 1
_SMALL_     equ 2
_MEDIUM_    equ 3
_COMPACT_   equ 4
_LARGE_     equ 5
_HUGE_      equ 6
DATE_MARI  equ (@MODEL EQ COMPACT) OR (@MODEL EQ LARGE)
COD_MARE   equ (@MODEL EQ MEDIUM) OR (@MODEL EQ LARGE)
DATE_MICI  equ NOT (DATE_MARI)
COD_MIC    equ NOT (COD_MARE)
sablon struc
    _bp    dw    ?
    _ip    dw    ?
IF COD_MARE
    _cs    dw    ?
```

```
ENDIF
IF DATE_MICI
    adr_var      dw ?
ELSE
    adr_var      dd ?
ENDIF
    sablon ends

    portabila proc
        push  bp
        mov   bp, sp
    IF DATE_MARI
        les   bx, [bp].adr_var
    ELSE
        mov   bx, [bp].adr_var
    ENDIF
        pop   bp
        ret
    portabila endp
```

Şablonul de acces se modifică atât la schimbarea tipului datelor cât și la cea a codului. Similar, accesul la `adr_var` presupune încărcarea offset-ului sau a perchii (segment:offset).

Segmentele definite care au importanță din punct de vedere ASM sunt următoarele:

_TEXT sau SURSA_TEXT

Corespund directivei `.code` din ASM sau codului generat de compilatorul C. La modelele de cod mare (`compact`, `large`, `huge`) se generează numele `SURSA_TEXT`, unde `SURSA` este numele fișierului sursă C sau ASM. La ASM se poate folosi directiva `.code nume`, caz în care se va genera un segment cu numele `nume_TEXT`. La modelele de cod mic (`tiny`, `small`, `compact`), se generează segmente cu numele `_TEXT`, cu tipul de combinare `PUBLIC`, ceea ce va face ca toate segmentele `_TEXT` să se combine într-unul singur, rezultând astfel un unic segment de cod.

_DATA sau SURSA_DATA

Corespund directivei `.data` din ASM sau datelor din C în clasa extern și static internal (deci cele cu alocare statică) inițializate explicit. La modelul `huge`, compilatorul C generează numele `SURSA_DATA`, unde `SURSA` este numele fișierului sursă, iar la celelalte, numele `_DATA`. Tipul de combinare este `PUBLIC`, deci toate segmentele cu numele `_DATA` se vor combina într-unul singur.

_BSS

Corespunde directivei `.data?` din ASM sau datelor cu alocare statică neinițializate explicit. Acestea vor fi inițializate de modulul `c0` cu valoarea 0. La modelul `huge`, atât datele neinițializate cât și cele inițializate sunt definite în segmentele `SURSA_DATA`. Tipul de combinare este `PUBLIC`, deci toate segmentele cu numele `_BSS` se vor combina într-unul singur.

_STACK

Corespunde stivei programului. Stiva este declarată cu o lungime minimă în `c0`, dar poate fi extinsă în ASM prin directiva `.stack`, sau în C prin declarația:

```
extern unsigned _stklen = ... ;
```

La modelele diferite de huge și tiny, se generează un grup de segmente numit DGROUP, care cuprinde segmentele _DATA și _BSS (între altele). La modelul tiny acest grup cuprinde și segmentele _TEXT și _STACK. La modelul huge, DGROUP nu cuprinde segmentele SURSA_DATA.

Compilatorul C generează următoarele segmente, funcție de modelul de memorie folosit:

- **tiny**

Se va genera un unic segment de cod, cu numele _TEXT, un unic segment de date inițializate _DATA și un unic segment de date neinițializate _BSS. Se generează un grup de segmente, numit DGROUP, care va include segmentele _TEXT, _DATA și _BSS. Nu se definește segment de stivă. Acest model corespunde programelor de tip .COM din DOS. În ASM se vor folosi directivele .code pentru cod, .data pentru date inițializate și .data? pentru date neinițializate. Registrul DS este inițializat cu DGROUP. Toți pointerii sunt de tip near. Apelurile de funcții sunt implicit near. Modulul de inițializare este c0t.obj, iar bibliotecile specifice sunt cs.lib și maths.lib.

- **small**

Se va genera un unic segment de cod, cu numele _TEXT și un grup de segmente, numit DGROUP, care va include segmentul de date inițializate _DATA și segmentul de date neinițializate _BSS. Se definește explicit un segment de stivă. Toți pointerii sunt implicit de tip near. Apelurile de funcții sunt implicit de tip near. Registrul DS este inițializat în modulul c0 cu DGROUP. Modulul de inițializare specific este c0s.obj, iar bibliotecile specifice sunt cs.lib și maths.lib.

- **compact**

Se va genera un unic segment de cod, cu numele _TEXT și un grup de segmente de date, numit DGROUP care conține segmentele _DATA (accesibil în ASM prin directiva .data) și _BSS (accesibil în ASM prin .data?). Pointerii la funcții sunt implicit de tip near, iar pointerii la date sunt implicit de tip far. Apelurile de funcții sunt implicit de tip near. Registrul DS este inițializat în modulul c0 cu DGROUP. Modulul de inițializare specific este c0c.obj, iar bibliotecile specifice sunt cc.lib și mathc.lib.

- **medium**

Se vor genera mai multe segmente de cod, cu numele SURSA_TEXT, unde SURSA este numele fișierului sursă C sau ASM. În ASM, dacă directiva code are un parametru, se va genera un segment de cod cu numele respectiv. Se consideră că în cuprinsul segmentului de cod respectiv, este activă o directivă ASSUME CS: cu numele segmentului respectiv. Se generează grupul DGROUP care include segmentele _DATA și _BSS (accesibile în ASM prin .data și .data?). Pointerii la date sunt implicit de tip near, iar pointerii la funcții sunt implicit de tip far. Apelurile de funcții sunt implicit de tip far. Registrul DS este inițializat în modulul c0 cu DGROUP. Modulul de inițializare specific este c0m.obj, iar bibliotecile specifice sunt cm.lib și mathm.lib.

- **large**

Se vor genera mai multe segmente de cod, cu numele SURSA_TEXT, unde SURSA este numele fișierului sursă C sau ASM. În ASM, dacă directiva code are un parametru, se va genera un segment de cod cu numele respectiv. Se consideră că în cuprinsul segmentului de cod respectiv, este activă o directivă ASSUME CS: cu numele segmentului respectiv. Se generează grupul DGROUP care include segmentele _DATA și _BSS (accesibile în ASM prin .data și .data?). Toți pointerii sunt implicit de tip far. Apelurile de funcții sunt implicit de tip far. Registrul DS este inițializat în modulul c0 cu DGROUP. Modulul de inițializare specific este c0l.obj, iar bibliotecile specifice sunt cl.lib și mathl.lib.

- **huge**

Se vor genera mai multe segmente de cod, cu numele SURSA_TEXT, unde SURSA este numele fișierului sursă C sau ASM. În ASM, dacă directiva code are un parametru, se va genera un segment de cod cu numele respectiv. Se vor genera mai multe segmente de date, cu numele SURSA_DATA, unde SURSA este numele fișierului sursă C sau ASM. Nu se generează segmente _BSS, toate datele fiind definite în segmentele SURSA_DATA. Se consideră că în cuprinsul modulului respectiv este activă o directivă ASSUME CS:SURSA_TEXT, DS:SURSA_DATA. Grupul de segmente DGROUP nu include segmentele SURSA_DATA. Asmblorul TASM generează însă un segment cu numele _DATA, corespunzător directivei .data, care se include în grupul DGROUP. Registrul DS este inițializat în modulul c0 cu DGROUP iar la fiecare intrare într-o funcție C (inclusiv main), registrul DS este inițializat cu SURSA_DATA. Evident, la intrarea în funcțiile ASM, va trebui făcută o inițializare cu DGROUP, salvând în prealabil registrul DS în stivă și refăcându-l la ieșirea din funcția ASM. Pentru accesul la segmentele de date definite în C se poate folosi operatorul SEG aplicat unei variabile extern, sau direct numele segmentului, în forma SURSA_DATA.

Toți pointerii sunt implicit de tip far și sunt normalizați, adică offsetul este redus la intervalul de valori 0...15. Astfel, corespondența segment:offset cu adresa fizică de memorie devine biunivocă. Incrementarea/decrementarea pointerilor se va reflecta acum atât în offset cât și în adresă de segment, spre deosebire de modelele celelalte, la care incrementarea/decrementarea pointerilor de tip far acționează numai asupra offsetului. Este posibilă deci definirea unor structuri de date de dimensiuni mai mari decât 64K, ceea ce la celelalte modele de memorie nu este posibil.

Apelurile de funcții sunt implicit de tip far. Modulul de inițializare specific este c0h.obj, iar bibliotecile specifice sunt ch.lib și mathh.lib.

O problemă specială poate apărea la modelele de memorie așa-zise "de date mici" (small și medium). Se pot defini date de orice fel, atât în clasa de alocare automatic (cu spațiu rezervat în stivă, deci accesibile prin registrul de segment SS), cât și în clasa de alocare static (cu spațiu rezervat într-unul din segmentele ce compun grupul DGROUP, deci accesibile prin registrul de segment DS).

În aceste modele de memorie, pointerii către date sunt de tip near, deci ei memorează numai deplasamentele obiectelor în cadrul segmentelor respective (de date, pentru obiecte în clasa static, respectiv de stivă, pentru obiecte în clasa auto). Dacă se accesează obiecte (date) prin pointeri, nu se poate ști în

ce tip de segmente sunt definite aceste obiecte. Problema este rezolvată (în cazul modelelor small și medium) prin generarea segmentelor de aşa manieră, încât adresele de început ale segmentului de stivă și grupului DGROUP să coincidă. Practic, pe parcursul execuției, registrele DS și SS vor avea aceeași valoare.

Să observăm că problema nu se pune în cazul modelului tiny (există un unic segment) și nici în cazul modelelor large și huge (toți pointerii sunt de tip far). Deasemenea, problema nu se pune în cazul modelului "de cod mic și date mari" compact, deoarece pointerii către funcții vor conține (indiferent de locul unde sunt definiți) deplasamente în cadrul unicului segment de cod care se generează. Apelurile de funcții prin pointeri se traduc în simple salturi indirecte intrasegment (instructiuni de tip call word ptr ...). Exemple de asemenea apeluri sunt:

```
call    word ptr [bp-4]      ;Apel indirect, pointer în clasa auto  
call    word ptr _p         ;Apel indirect, pointer în clasa static
```

Dacă se fac indirectări multiple, acestea presupun definirea unor pointeri către pointeri către funcții, adică a unor pointeri către date. Acești pointeri vor fi deci de tip far și vor fi încărcați și dereferențiați totdeauna ca adrese complete (segment și offset).

În concluzie, să reținem că, la modelele small și medium, registrele DS și SS vor avea același conținut la execuție. Mediul integrat oferă opțiunea **Options/Compiler/Code Generation/Assume DS not equal SS**, care este recomandabil să fie plasată pe **Default for memory model**.

Pe lîngă segmentele descrise mai sus, modulul de inițializare c0 mai definește o serie de segmente, folosite la inițializare, la ieșirea în sistemul de operare, etc, dar care nu prezintă interes din punct de vedere ASM. Se poate vedea exact ce segmente se generează, studiind fișierul sursă c0.asm, furnizat în kit-ul implementării C (în directorul \examples\startup).

Când există îndoieri asupra a ce se generează concret de către compilatorul C, se poate folosi opțiunea -S la compilare în linia de comandă, obținându-se în acest fel un fișier sursă ASM, corespunzător modulului C respectiv sau se poate lansa Turbo Profiler-ul din mediu integrat, care, între alte informații, furnizează și codul ASM generat.

7.4 O aplicație mixtă cu tipuri de date simple

Pentru a fixa ideile, considerăm un exemplu de program, compus dintr-un modul C și un modul ASM, care va fi detaliat în trei cazuri de modele: compact, medium și huge. Codul ASM pentru celelalte modele se deduce simplu: la modelul small se tratează datele ca la medium și codul ca la compact, iar la modelul large se tratează datele ca la compact și codul ca la medium.

Modulul sursă C, presupus în fișierul test1c.c, este același în toate situațiile, și anume:

```
#include <stdio.h>  
extern int f (int);  
extern int (*pf) (int);  
extern int *g (void);  
extern int *gg (void);  
extern int ext_i;
```

```
extern int ext_n;
int extc_i = 2;

int *ff (int *pi)
{
    *pi *= 2;
    return pi;
}
void main(void)
{
    int n;
    printf("\nIntroduceti un intreg: ");
    scanf("%d",&n);
    n = f(n);
    printf("Valoarea dubla este: %d\n",n);
    n = (*pf)(n);
    printf("Valoarea dubla este: %d\n",n);
    printf("Variabila externa initializata este: %d\n", ext_i);
    ext_n = 5;
    printf("Variabila externa neinitializata este: %d\n", *g());
    printf("Valoarea dubla este: %d\n", *gg());
}
```

Modulele ASM sunt presupuse în fișierele test1ca.asm, test1ma.asm și, respectiv, test1ha.asm, corespunzătoare modelelor compact, medium și huge.

În aplicația studiată se consideră următoarele entități de program definite în ASM:

- funcția f, cu tipul int și cu un parametru de tip int;
- pointerul la funcție pf, inițializat cu adresa funcției f;
- funcția g, cu tipul int și fără parametri;
- funcția gg, cu tipul int * și fără parametri;
- întregul ext_i, inițializat cu valoarea 11;
- întregul ext_n, neinițializat.

În modulul C se consideră următoarele entități:

- întregul extc_i, inițializat cu valoarea 2;
- funcția ff, cu tipul int * și cu un parametru de tip int *.

Funcția ff primește o adresă de întreg și înmulțește cu 2 întregul de la adresa respectivă, întorcând adresa primită.

Funcția f primește un întreg, afișează la consolă mesajul "Intregul este:" și valoarea variabilei extc_i. Se întoarce programului apelant valoarea dublă a parametrului primit.

Funcția g întoarce programului apelant adresa variabilei ext_n.

Funcția gg apelează funcția ff, cu parametrul adresa variabilei ext_n, întorcând programului apelant adresa returnată de de funcția ff, adică adresa aceeași variabilă n.

Programul principal, scris în C, citește de la consolă un întreg n și apelează de două ori funcția f, cu parametrul n, prima dată direct iar a doua oară prin pointerul pf. În ambele apeluri, valoarea întoarsă este depusă în aceeași variabilă n. Se afișează apoi variabila ext_i, se modifică explicit variabila ext_n

și se apelează funcțiile g și gg, afișând la întregii de la adresele întorse de aceste funcții, adică variabila ext_n.

Afișările din modulul ASM sunt realizate prin macroinstructiunile putsi și puti, care presupun datele adresabile prin registrul DS. Rezultă deci că aplicația va trebui să cuprindă și modulul obiect io.obj (vezi Anexa B).

7.4.1 Dezvoltare în modelul de memorie compact

Modulul test1ca.asm, corespunzător modelului compact, este:

```
.model compact
include io.h
public _f, _pf, _g, _gg, _ext_i, _ext_n
extrn _ff: near, _extc_i: word
sablon_f struc
    _bp    dw ?
    _ip    dw ?
    n      dw ?
sablon_f ends
.data
    _ext_i dw     11
    _pf    dw     _f
    ; Întreg inițializat
    ; Pointer near inițializat cu
    ; adresa (offset-ul) lui _f
    ; Întreg neinițializat
.data?
    _ext_n dw     ?
.code
_f proc near
    push  bp
    mov   bp, sp
    mov   ax, [bp].n
    add   ax, ax
    putsi <cr,lf,'Întregul este: '>
    mov   bx, DGROUP:_extc_i
    puti  bx
    putsi <cr, lf>
    pop   bp
    ret
_f endp

_g proc near
    mov   dx, DGROUP
    lea   ax, DGROUP:_ext_n
    ret
_g endp

_gg proc near
    mov   ax, DGROUP
    push  ax
    lea   ax,DGROUP:_ext_n
    push  ax
    call  near ptr _ff
    add   sp, 4
    ret
_gg endp
end
```

; Poziționare bp pentru acces
; Preluare parametru
; Dublare
; Segmentul este DGROUP
; Tipărește întregul (bx)
; (ax) = rezultat
; Segment
; Offset
; Rezultat in (dx:ax)
; Pregătire parametri apel
; Segment
; Offset
; Apel _ff
; Descărcare stiva
; Rezultat in (dx:ax)

Deoarece modelul de memorie este compact, toate apelurile de funcții sunt de tip near, pointerul pf la funcție este de tip near, iar toți pointerii la date sunt de

tip far. Funcțiile f, g și gg vehiculează adrese de 32 de biți, transmise prin stivă sau întoarse prin perechea (DX:AX).

Programul va afișa la consolă textul:

```
Introduceti un intreg: 13
Intregul C este: 2
Valoarea dubla este: 26
Intregul C este: 2
Valoarea dubla este: 52
Variabila externă initializată este: 11
Variabila externă neinitializată este: 5
Valoarea dubla este: 10
```

Presupunând directorul mediului C ca fiind C:\BC, secvența de linii de comandă pentru dezvoltarea aplicației este:

```
bcc -mc -c -Ic:\bc\include test1.c
tasm test1ca.asm/ml
tlink c:\bc\lib\c0c test1c test1ca io,test1c,,c:\bc\lib\cc
```

7.4.2 Dezvoltare în modelul de memorie medium

Modulul test1ma.asm, corespunzător modelului de memorie medium este:

```
.model medium
include io.h
public _f, _pf, _g, _gg, _ext_i, _ext_n
extrn _ff: far, _extc_i :word
sablon_f struc
    _bp    dw ?
    _ip_cs dd ?
    n      dw ?
sablon_f ends
.data
    _ext_i dw 11
    _pf    dd _f
    ; Întreg inițializat
    ; Pointer far inițializat cu
    ; adresa completă a lui _f
    ; Întreg neinițializat

.data?
    _ext_n dw ?

.code
_f    proc far
    push  bp
    mov   bp, sp
    mov   ax, [bp].n
    add   ax, ax
    mov   bx, DGROUP:_extc_i
    putsi <cr, lf,'Intregul C este: '>
    puti  bx
    putsi <cr, lf>
    pop   bp
    retf
    ; (ax) = rezultat
_f    endp

_g    proc far
    lea   ax,DGROUP:_ext_n
    retf
    ; Offset
    ; Rezultat în (ax)
_g    endp

_gg   proc far
    lea   ax, DGROUP:_ext_n
    push  ax
    ; Pregătire parametri apel
    ; Offset
```

```

        call    far ptr _ff           ; Apel _ff
        add     sp, 2                ; Descărcare stiva
        retf
_gg      endp
end

```

Ceea ce se schimbă față de modelul anterior este faptul că apelurile și pointerii la funcții sunt de tip far, iar adresele și pointerii la variabile sunt de tip near. Şablonul de acces în stivă se modifică, ținând cont că un apel far salvează și registrul CS în stivă. Pointerul pf se definește cu directiva dd (define doubleword). Adresa lui ext_n (offset) este întoarsă de funcțiile g și gg prin AX.

Secvența de linii de comandă este:

```

bcc -mm -c -Ic:\bc\include test1c.c
tasm test1ma.asm/ml
tlink c:\bc\lib\c0m test1c test1ma io,test1m,,c:\bc\lib\cm

```

7.4.3 Dezvoltare în modelul de memorie huge

Modulul test1ha.asm, corespunzător modelului huge este:

```

.model huge
include io.h
public _f, _pf, _g, _gg, _ext_i, _ext_n
extrn _ff:far, _extc_i:word

sablon_f struc
    _bp    dw ?
    _ip_cs dd ?
    n      dw ?
sablon_f ends

.data
    _ext_i dw 11          ; Întreg inițializat
    _pf    dd _f          ; Pointer far inițializat cu
                          ; adresa completă a lui _f

.data?
    _ext_n dw ?          ; Întreg neinițializat

code
_f proc far
    push   bp
    mov    bp,sp          ; Poziționare bp pentru acces
    push   ds
    push   es
    push   bx
    mov    ax,DGROUP
    mov    ds,ax          ; Segment local de date

    mov    ax,[bp].n
    add    ax,ax          ; Preluare parametru
                          ; Dublare

    push   ax
    mov    ax,SEG _extc_i
    mov    es,ax
    mov    bx,es:_extc_i  ; Preluare _extc_i
;

; Sirul din macroinstructiunea putsi se definește într-o
; directivă .data, deci în segmentul local, de aceea ds
; trebuie să indice DGROUP

```

```

;
putsi <cr,lf,'Intregul C este: '>
puti  bx ;Tipărește întregul (bx)
putsi <cr,lf>
pop   ax ; Rexultat

pop   bx ; Refacere bx
pop   es ; Refacere es
pop   ds ; Refacere ds

pop   bp ; (ax) = rezultat
_f endp

_g proc far
    mov  dx,DGROUP ; Segment
    lea   ax,DGROUP:_ext_n ; Offset
    retf ; Rezultat în (dx:ax)
_g endp

_gg proc far
    mov  ax,DGROUP ; Pregătire parametri
    push ax ; Segment local
    lea   ax,DGROUP:_ext_n
    push ax ; Offset
    call  far ptr _ff ; Apel _ff
    add   sp,4 ; Descărcare stiva
    retf ; Rezultat în (dx:ax)
_gg endp
end

```

Se observă că, în procedura f, unde se face acces la date, se poziționează explicit registrul DS pe grupul de segment DGROUP, pentru a putea utiliza macroinstructiunile de afișare și se încarcă explicit în ES adresa de segment a variabilei externe ext_c. Registrele DS și ES se salvează la intrarea în funcție și se refac înainte de întoarcerea în programul apelant.

Secvența de linii de comandă este:

```

bcc -mh -c -Ic:\bc\include test1c.c
tasm test1ha.asm/ml
tlink c:\bc\lib\c0h test1c test1ha io,test1h,,c:\bc\lib\ch

```

7.5 O aplicație mixtă cu tipuri de date structurate

Se consideră modulul sursă C, presupus în fișierul test2c.c:

```

#include <string.h>
struct tips { int n; char *sir; };
void f (struct tips *);
struct tips *g (struct tips *);
struct tips * (*pg) (struct tips *) = g;

struct tips a = { 100, "Iata un sir..." };

struct tips *g(struct tips *x)
{
    static int contor = 0;
    contor++;
    f(x);
    (x -> n)++;
}

```

```

        if(contor % 2)
            strupr(x -> sir);
        else
            strlwr(x -> sir);
        f(x);
        return x;
    }

```

Se consideră tipul de structură tips, care cuprinde un întreg și un pointer la un sir de caractere. Structura a este inițializată cu valoarea 100, respectiv cu sirul "Iata un sir...".

Funcția de tip void f, scrisă în ASM, primește o adresă de structură și afișează la consolă cele două câmpuri (întreg și sir

de caractere) cu ajutorul macroinstructiunilor piuti și puts, definite în fișierul header io.h.

Funcția g, scrisă în C, primește adresa x a unei structuri și realizează următoarele operații:

- apelul funcției f, cu parametrul adresa x (care va afișa structura indicată de x);
- incrementarea câmpului numeric al structurii de la adresa x;
- modificarea succesivă a sirului de caractere din structura indicată de variabila x, din litere mici în litere mari și succesiv; acest lucru este realizat prin incrementarea modulo 2 a unei variabile locale statice;
- apelul funcției f, cu parametrul adresa x (care va afișa structura indicată de x).

Programul principal, scris în ASM, realizează următoarele operații:

- apeleză funcția g în mod direct, cu parametrul adresa structurii a;
- pregătește în stivă adresa întoarsă de g (deci adresa structurii a), pentru un apel viitor;
- apeleză funcția f în mod direct, cu parametrul întors de apelul lui g (deci cu adresa structurii a);
- apeleză indirect funcția g, prin intermediul pointerului pg, cu parametrul adresa pregătită anterior în stivă;
- apeleză funcția f, cu parametrul întors de apelul indirect al lui g (deci cu adresa structurii a).

7.5.1 Dezvoltare în modelul de memorie compact

Modulul test2ca.asm, corespunzător modelului de memorie compact, este:

```

.model      compact
include    io.h
public     _f, _main
extrn      _g: near, _a: far, _pg: word

tips      struc
    n      dw ?
    sir    dd ?          ; Şablon pentru
                        ; structura tips
tips      ends

sablon struc
    _bp     dw ?          ; Şablon pentru acces
    _ip     dw ?          ; în stivă
    _adr_struc dd ?      ; Parametrul
sablon ends

.stack 1024

```

```
.code

_f    proc near
      push  bp
      mov   bp, sp
      push  ds          ; Salvări
      push  di          ; registre
      push  si          ; folosite

;
; Încărcare adresă far din stivă
;
      les   di, dword ptr [bp]._adr_struc
;
; Acces la membrul n al structurii
;
      mov   ax, es:[di].n
;
; Tipărire n
;
      putsi <cr, lf, 'Intregul este: '>
      puti  ax
      putsi <cr, lf, 'Sirul este: '>
;
; Acces la membrul sir (pointer far) al structurii
;
      lds   si, dword ptr es:[di].sir
;
; Tipărire sir
;
      call  far ptr puts_proc
      pop   si          ; Refaceri
      pop   di          ; registre
      pop   ds          ; folosite
      pop   bp
      ret

_f endp

_main proc near
      mov   ax,DGROUP           ; Pregătire parametri
      push  ax
      lea   ax,DGROUP:_a
      push  ax
      call  near ptr _g
      add   sp,4               ; Descărcare stivă
;
; g întoarce un pointer far în (dx:ax)
; Este pregătit acest pointer în stivă, pentru un apel viitor al lui g
; (altfel ar fi trebuit salvat într-o zonă de date)
;
      push  dx          ; Segment
      push  ax          ; Offset
;
; Pregătire parametru pentru apelul lui f
;
      push  dx          ; Segment
      push  ax          ; Offset
      call  near ptr _f
      add   sp, 4        ; Descărcare stivă
```

```

;
; Acum se face apelul lui g, indirect, prin pointerul de tip near pg,
; cu parametrul de tip pointer far, pregătit anterior în stivă
;
    call    word ptr _pg
    add    sp, 4           ; Descărcare stivă
;
; g întoarce un pointer far în (dx:ax)
; Se pregătește acest pointer în stivă, pentru apelul lui f
;
    push   dx             ; Segment
    push   ax             ; Offset
    call    near ptr _f
    add    sp, 4           ; Desccărcare stivă
    ret                 ; Terminare program
_main endp
end

```

Pointerii la date sunt de tip far iar pointerul lica funcție este de tip near. Se observă declararea simbolului extern a ca o etichetă de tip far, ceea ce este în concordanță cu convenția că numele unei variabile este sinonim cu adresa variabilei respective.

La consolă se va tipări:

```

Intregul este: 100
Sirul este: Iata un sir...
Intregul este: 101
Sirul este: IATA UN SIR...
Intregul este: 101
Sirul este: IATA UN SIR...
Intregul este: 101
Sirul este: IATA UN SIR...
Intregul este: 102
Sirul este: iata un sir...
Intregul este: 102
Sirul este: iata un sir...

```

Secvența de linii de comandă este:

```

bcc -mc -c -Ic:\bc\include test2c.c
tasm test2ca.asm/ml
tlink c:\bc\lib\c0c test2ca test2c io,test2c,,c:\bc\lib\cc

```

7.5.2 Dezvoltare în modelul de memorie medium

Modulul test2ma.asm, corespunzător modelului medium este:

```

.model    medium
include  io.h
public   _f, _main
extrn   _g:far, _a:near, _pg:dword
tips struc
    n      dw ?          ; Şablon pentru
    sir    dw ?          ; structura tips
tips ends

sablon struc
    _bp        dw ?      ; Şablon pentru acces
    _ip_cs    dd ?      ; în stivă
    _adr_struc dw ?      ; Parametrul
sablon ends

```

```

.stack 1024
.code

_f proc far
    push  bp
    mov   bp, sp
    push  di          ; Registre
    push  si          ; folosite
;
    mov   di,word ptr [bp]._adr_struc
;
    mov   ax,DGROUP:[di].n
;
    putsi <cr,lf,'Intregul este: '>
    puti  ax
    putsi <cr,lf,'Sirul este: '>
;
    mov   si,word ptr DGROUP:[di].sir
;
    call  far ptr puts_proc
    pop   si          ; Refaceri
    pop   di          ; registre
    pop   bp
    retf
_f endp

_main proc far
    lea   ax,DGROUP:_a          ; Pregătire parametru
    push  ax
    call  far ptr _g            ; Offsetul structurii a
    add   sp,2                  ; Apel g
                                ; Descărcare stivă
;
; g întoarce un pointer near în (ax)
;
    push  ax          ;Offset
;
; Pregătire parametru pentru apelul lui f
;
    push  ax          ; Offset
    call  far ptr _f            ; Apel f
    add   sp,2          ; Descărcare stivă
;
    call  dword ptr _pg          ; Descărcare stivă
    add   sp, 2
;
; g întoarce un pointer near în (ax)
;
    push  ax          ;Offset
    call  far ptr _f            ; Apel f
    add   sp, 2          ; Descărcare stivă
    retf
_main endp
end

```

Diferențele față de cazul precedent rezultă din faptul că funcțiile și pointerii la funcții sunt de tip far, iar adresele variabilelor și pointerii la date sunt de tip near.

Simbolul extern **a** (de tip structură) se declară în ASM ca o etichetă de tip near.

Secvența de linii de comandă este:

```
bcc -mm -c -Ic:\bc\include test2c.c  
tasm test2ma.asm/ml  
tlink c:\bc\lib\c0m test2ma test2c io,test2m,,c:\bc\lib\cm
```

7.5.3 Dezvoltare în modelul de memorie huge

Modulul test2ha.asm, corespunzător modelului huge este:

```
.model      huge  
public      _f, _main  
extrn      _g:far, _a: far, _pg:dword  
include    io.h  
  
tips struc  
    n      dw ?           ; Şablon pentru  
    sir    dd ?           ; structura tips  
tips ends  
  
sablon struc  
    _bp     dw ?           ; Şablon pentru acces  
    _ip_cs  dd ?           ; în stivă  
    _adr_struc dd ?       ; Parametru  
sablon ends  
  
.stack 1024  
.code  
  
_f proc far  
    push  bp  
    mov   bp, sp  
    push  ds           ; Salvări  
    push  di           ; registre  
    push  si           ; folosite  
;  
    mov   ax, DGROUP  
    mov   ds, ax  
;  
    les   di,dword ptr [bp+6]  
    mov   ax,es:[di].n  
    putsi <cr,lf,'Intregul este: '>  
    puti  ax  
    putsi <cr,lf,'Sirul este: '>  
    lds   si,dword ptr es:[di].sir  
    call  far ptr puts_proc  
;  
    pop   si           ; Refaceri  
    pop   di           ; registre  
    pop   ds           ; folosite  
    pop   bp  
    retf  
_f endp  
  
_main proc far  
    push  ds  
    push  es  
    mov   ax,DGROUP          ; Comutare pe segmentul
```

Gheorghe Muscă – Programare în Limbaj de Asamblare

```
        mov    ds,ax           ; propriu de date
;
        mov    ax, SEG _a      ; Pregătire parametru
        mov    es, ax           ; Segmentul structurii a
        push   es
        lea    ax, es:_a       ; Offsetul structurii a
        push   ax
        call   far ptr _g      ; Apel g
        add    sp,4             ; Descărcare stivă
;
; g întoarce un pointer far în (dx:ax)
; Este pregătit acest pointer în stivă, pentru
; un apel viitor al lui g (altfel ar fi trebuit salvat
; într-o zonă de date)
;
        push   dx           ; Segment
        push   ax           ; Offset
;
; Pregătire parametru pentru apelul lui f
;
        push   dx           ; Segment
        push   ax           ; Offset
        call   far ptr _f      ; Apel f
        add    sp,4             ; Descărcare stivă
;
; Acum se face apelul lui g, indirect, prin pointerul
; de tip near pg (care e în alt segment de date), cu
; parametrul de tip pointer far, pregătit anterior în stivă
;
        mov    ax, SEG _pg
        mov    es,ax
        call   dword ptr es:_pg
        add    sp,4
;
; g întoarce un pointer far în (dx:ax)
; Se pregătește acest pointer în stivă, pentru apelul lui f
;
        push   dx
        push   ax
        call   far ptr _f
        add    sp,4
;
        pop    es
        pop    ds
        retf
_main endp
end
```

Diferența față de celelalte cazuri este că datele externe nu mai sunt în mod necesar definite în același segment (grup) cu datele din modulul ASM. Adresarea structurii a și a pointerului pg trebuie făcută în consecință, folosind operatorul SEG pentru luarea segmentelor în care acești simboli sunt definiți.

Secvența de linii de comandă este:

```
bcc -mh -c -Ic:\bc\include test2c.c
tasm test2ha.asm/ml
tlink c:\bc\lib\c0h test2ha test2c io,test2h,,c:\bc\lib\ch
```

7.6 Dezvoltarea într-un limbaj mixt (C și ASM)

Mediul de dezvoltarea Borland C oferă posibilitatea inserării de cod ASM în programe sursă C. Fiecare instrucțiune ASM trebuie precedată de cuvântul cheie `asm` sau să fie inclusă într-un bloc de tip ASM, de forma:

```
asm {
    ; Instrucțiuni ASM
}
```

Acest mod de dezvoltare este foarte comod, deoarece vizibilitatea obiectelor definite în C se extinde și asupra blocurilor ASM. Astfel, putem folosi nume de variabile, nume de parametri formali ai funcțiilor etc., fără a mai scrie caracterul `_` în fața numelor respective. Se evită astfel și secvențele de intrare și revenire din funcții, care sunt realizate automat de compilatorul C.

O altă facilitate este accesul direct din C asupra registrelor procesorului, realizat **prin variabilele C predefinite `_AX`, `_BX`, `_CX`** etc., care pot fi utilizate, de exemplu, în instrucțiuni de atribuire.

Există restricții de utilizare asupra etichetelor. Mai precis, etichetele trebuie definite ca etichete C, deci în afara blocurilor ASM.

Ca și la dezvoltarea în module separate, programatorul trebuie să gestioneze explicit modelele de memorie, în sensul definirii precise a adreselor far sau `near`. Pentru aceasta, se utilizează modelele de memorie C și/sau cuvintele cheie `near` și `far` ale implementării Borland C.

Un modul dezvoltat în această manieră este scris de fapt într-un limbaj mixt (C și ASM). De altfel, fișierele sursă de acest tip se creează de obicei cu extensia `.CAS` (de la C și ASM).

Pentru exemplificare, să considerăm o funcție de căutare liniară într-un tablou de întregi. În limbajul C, am scrie această funcție în forma:

```
int cauta (int *a, size_t n, int x)
{
    int i;
    for (i = 0; i < n; i++)
        if (x == a[i])
            return i; /* S-a găsit întregul x */
    return -1; /* Nu s-a găsit întregul x */
}
```

Dorim avut să scriem aceeași funcție în "limbajul CAS". Presupunem pentru început un model "de date mici". Implementarea funcției este următoarea:

```
int cauta (int near *a, size_t n, int x)
{
    asm {
        push si ; Salvări
        push cx ; registre
        push dx

        mov si, a ; Adresa near a tabloului
        mov cx, n ; Număr elemente
        jcxz not_gasit ; Test n == 0
        mov dx, x ; Element căutat
        sub si, 2 ; Pornim cu o poziție mai la stânga
    }
}
```

reluare:

```

asm {
    add    si, 2           ; Element curent din tablou
    cmp    [si], dx        ; Comparăm cu cel căutat
    loopnz reluare        ; Nu este, se reia bucla
    jnz    not_gasit       ; Am ieșit din buclă
    mov    ax, n            ; Dacă ZF a fost 1, calculăm
    sub    ax, cx            ; indicele elementului găsit
    dec    ax                ; ca fiind n - CX - 1
    jmp    final             ; Salt la ieșire
}
not_gasit:
asm    mov    ax, -1          ; Nu s-a găsit elementul x
final:
asm {
    pop    dx                ; Refaceri
    pop    cx                ; registre
    pop    si
}
return _AX;                      ; Instrucțiune C
}

```

Se observă tehnica de scriere a buclei de căutare (prefixul LOOPNZ) și calculul indicelui elementului găsit, ca diferență dintre dimensiunea *n* a tabloului și valoarea incrementată a contorului CX la ieșirea din buclă. Revenirea în funcția apelantă se face cu instrucțiunea C return _AX (ne amintim convenția de întoarcere a datelor simple).

Un punct important este încărcarea adresei tabloului. Dacă adresa este near, o încărcăm cu o instrucțiune MOV (în stivă se găsește deplasamentul în cadrul segmentului curent de date). În cazul unui model de "date mari", în stivă se găsește un pointer far (o variabilă de tip DoubleWord), caz în care încărcarea s-ar face cu o instrucțiune:

```
les    si, a              ; Adresa far a tabloului
```

Compararea elementului curent cu cel căutat s-ar scrie:

```
cmp    es:[si], dx        ; Comparăm cu cel căutat
```

În fine, ar mai trebui salvat și restaurat registrul ES, iar prototipul funcției ar trebui scris în forma:

```
int cauta (int near *a, size_t n, int x);
```

Modificările de mai sus transformă transformă funcția într-una adecvată modelelor "de date mari".

Un program de test pentru funcția de mai sus s-ar scrie în genul:

```
#include <stdio.h>
int x [ ] = { 0, 1, 2, 3, 4, 5, 6, 7, 8, 9, };
#define NREL(a) (sizeof(a)/sizeof(a[0]))
void main (void) {
    int i, val = 6;
    i = cauta (x, NREL(x), val);
    if (i > 0)
        printf("Elementul %d se află pe pozitia %d din tablou\n", val, i);
    else
        printf("Elementul %d nu se află în tablou\n", val);
}
```

Trebuie totuși observat că această tehnică mixtă de dezvoltare are și dezavantaje. Un modul CAS este destul de greu de întreținut iar transportul său în alt mediu de dezvoltare decât Borland ar putea crea probleme de compatibilitate. Tehnica mixtă merită folosită atunci când avem de implementat secvențe relativ scurte de program, pentru care nu se justifică un modul (fișier sursă) separat.

7.7 Performanțe în ASM și C

În acest subcapitol vom face o analiză comparativă a unui algoritm implementat în C și ASM. Algoritmul ales este un algoritm de sortare internă, anume sortarea prin partiționare și interschimbare, numită și sortare rapidă (**quicksort**).

Algoritmul este recursiv și se poate implementa foarte comod atât în C cât și în ASM. O implementare polimorfică a acestui algoritm este oferită și de biblioteca standard a limbajului C.

7.7.1 Implementarea în C

Varianta C considerată va fi de asemenea polimorfică (capabilă să sorteze tablouri de orice fel). Prototipul funcției C este:

```
void qsort(void *tab, size_t n, size_t size, PFCMP cmp);
```

Semnificația parametrilor este:

- tab - adresa de început a tabloului;
- n - numărul de elemente al tabloului;
- size - dimensiunea în octeți a unui element;
- cmp - pointer la o funcție externă de comparație, definit prin:

```
typedef int (*PFCMP)(void *a, void *b);
```

Funcția de comparație (scrisă de utilizator) primește adresele a două elemente din tablou și întoarce o valoare negativă, zero sau pozitivă, după cum elementul de la prima adresă este mai mare, egal sau mai mic decât elementul de la a doua adresă. Sensul noțiunilor mai mare, egal sau mai mic este complet abstract, fiind definit de utilizator chiar prin funcția de comparație.

Pentru transferul datelor se vor utiliza funcțiile ajutătoare swap și copy, listate mai jos.

```
typedef unsigned char BYTE;

void swap(void *a, size_t size, int i, int j)
{
    /*
        Schimba a[i] cu a[j]. Elementele tabloului a[] au size octeti.

    */
    register BYTE *b = ((BYTE *)a) + i * size;
    register BYTE *c = ((BYTE *)a) + j * size;
    register BYTE temp;
    while(size--) {
        temp = *b;
        *b++ = *c;
        *c++ = temp;
    }
}
```

```

void copy(void *a, void *b, size_t size)
{
    /*
        Copiază obiectul de la adresa b în cel de la adresa a.
        Obiectele au size octeti.
    */
    memcpy(a, b, size);
}

```

Algoritmul de sortare rapidă este recursiv. Funcția care implementează algoritmul recursiv primește ca date de intrare o parte a tabloului care trebuie sortat, prin adresa de început și doi indici notați left și right. Inițial, funcția se apelează cu indicii 0 și n-1.

Se alege un element arbitrar al tabloului v, numit pivot, notându-l cu mark (variante uzuale sunt $v[\text{left}]$ sau $v[(\text{left}+\text{right})/2]$). Se partajează tabloul în raport cu mark, în sensul că toate elementele aflate la stînga lui mark să fie mai mici sau egale decît acesta, iar toate elementele aflate la dreapta lui mark să fie mai mari sau egale decît acesta.

În acest moment, pivotul se află pe poziția sa finală, iar tabloul este partionat în două subtablouri. Se apelează acum aceeași funcție, cu indicii left și k-1, respectiv k+1 și right, unde k este indicele pivotului în urma partionării. În felul acesta se sortează cele două subtablouri. Dacă $\text{left} \geq \text{right}$ algoritmul se oprește.

Algoritmul se poate îmbunătăți în felul următor. Să pornim cu doi indici i și j, inițializați cu left și, respectiv, cu right. Cît timp $v[i] < \text{mark}$, incrementăm i, apoi, cît timp $v[j] > \text{mark}$, decrementăm j. Dacă acum $i \leq j$, interschimbăm $v[i]$ cu $v[j]$, actualizînd similar indicii i și j. Tot acest proces continuă pînă cînd $i > j$. Acum se apelează recursiv funcția, cu indicii left și j, respectiv i și right (dacă $\text{left} < j$, respectiv dacă $i < \text{right}$).

În implementarea polimorfică, deoarece la compilare nu se cunoaște dimensiunea unei înregistrări, se alocă spațiu dinamic pentru înregistrarea mark, copiindu-se elementul din mijlocul tabloului și eliberînd spațul la ieșirea din funcție.

O îmbunătățire posibilă a metodei este recursarea la o metodă directă de sortare, dacă lungimea tabloului este inferioară unei

limite prestabilită, evitînd astfel apelul recursiv. De exemplu, la un tablou de 1000 de înregistrări, cînd dimensiunea partitiei ajunge 2, se vor face 500 de apeluri ale funcției, pentru a sorta de fiecare dată un tablou de 2 elemente. Este mult mai eficient dacă acest lucru se face direct. În implementarea de față, s-a ales limită 2, caz în care cele două înregistrări se compară direct. Pentru eficiență, variabilele intens folosite în etapa de partitîonare (i și j) sunt declarate în clasa register. Implementarea este următoarea:

```

void quick_sort(BYTE *v, size_t size, int left, int right, PFCMP cmp)
{
    register int i, j;
    BYTE *mark;

    i = left; j = right;
    switch(j - i) {
        case 0: return;

```

```

        case 1: if(cmp(v + i*size, v + j*size) > 0)
                  swap(v, size, i, j);
                  return;
        default: break;
    }
mark = (BYTE *) malloc(size);
copy(mark, v + ((left + right)/2)*size, size);
do {
    while(cmp(v + i*size, mark) < 0)
        i++;
    while(cmp(v + j*size, mark) > 0)
        j--;
    if(i <= j)
        swap(v, size, i++, j--);
} while (i <= j);
if(left < j)
    quick_sort(v, size, left, j, cmp);
if(i < right)
    quick_sort(v, size, i, right, cmp);
free(mark);
}

void Quick_C (void *v, size_t n, size_t size, PFCMP cmp)
{
    quick_sort(v, size, 0, (int) n-1, cmp);
}

```

7.7.2 Implementarea în ASM

Să considerăm acum implementarea algoritmului de sortare rapidă în limbaj de asamblare, într-o versiune specializată (pentru tablouri de întregi). Toate adresele se consideră de tip near, fiind offset-uri în cadrul segmentului adresat prin registrul DS.

Algoritmul este același cu cel prezentat în 7.7.1, implementarea ASM urmărind exact programul C. Variabilele din descrierea în C a algoritmului sunt menținute în registrele procesorului și în stivă.

Parametrii se transmit prin stivă, conform şablonului definit în program iar stiva este descărcată de programul apelant. Această modalitate de apel permite ca funcția ASM să poată fi apelată din C, dacă modelul cu care se lucrează este small.

```

.286
.model small
.code
public _qasm
;
;     qasm(int v[], int left, int right)
;
sablon      struc
    _bp_ip      dw 2 dup(?)
    v           dw ?
    left        dw ?
    right       dw ?
sablon ends

_qasm proc near
    push   bp
    mov    bp,sp

```

```

pusha
;
;   Asignarea variabilelor
;
;   i = si, j = di, v = bx, mark = dx
;   left, right = în stivă
;
mov  bx, [bp].v           ; v
mov  si, [bp].left        ; i = left
mov  di, [bp].right       ; j = right
mov  ax, di               ; compară
sub  ax, si               ; right - left
cmp  ax, 1                ; cu 1
jng  et00
jmp  et1                  ; mai mare: se executa
                           ; rutina normal
et00:
je   et0                  ; right = left + 1
jmp  gata                 ; left = right: iesire
et0:
;
; Sunt 2 elemente
; Se compară și eventual se schimbă
;
shl  si, 1                ; Întregii sunt pe
shl  di, 1                ; doi octeți
mov  ax, [bx][si]          ; v[left]
cmp  ax, [bx][di]          ; v[right]
jnle aici
jmp  gata                 ; Sunt O.K.
aici:
xchg ax, [bx][di]         ; Dacă nu, se
mov [bx][si], ax           ; schimbă
jmp gata                  ; și gata
et1:
;
; Sunt mai mult de două elemente
;
mov  ax, si                ; Calcul
add  ax, di                ; (left + right)/2
shr  ax, 1
push bx
shl  ax, 1                ; Adresa se adună
add  bx, ax                ; la v
mov  dx, [bx]              ; mark = v[(left+right)/2]
pop  bx
_do:
while_i:
;
shl  si, 1                ; Ciclu do
cmp  [bx][si], dx ; mark ; Ciclu while după
jge  end_i                ; v[i] < mark
                           ; Compară v[i] cu
                           ; Sfârșit ciclu dacă
                           ; v[i] >= mark
shr  si, 1
inc  si                   ; Dacă nu, se face i++
jmp  while_i              ; și se reia
end_i:
shr  si, 1

```

Gheorghe Muscă – Programare în Limbaj de Asamblare

```

while_j:                                ; Ciclu while după
    shl  di, 1                          ; v[j] > mark
    cmp  [bx][di], dx                  ; Compara v[j] cu mark
    jle  end_j                         ; Sfârșit ciclu dacă
                                         ; v[j] <= mark
    shr  di, 1                          ; Dacă nu, se face j--
    dec  di                            ; și se reia
    jmp  while_j
end_j:                                   
    shr  di, 1
;
    cmp  si, di                      ; Compara i cu j
    jg   _while                         ; Salt dacă i > j
;
    shl  di, 1                          ; Întregii sunt
    shl  si, 1                          ; pe doi octeți
    mov  ax, [bx][si]                  ; Schimbă v[i]
    xchg ax, [bx][di]                  ; cu
    mov  [bx][si], ax                  ; v[j]
    shr  si, 1
    shr  di, 1
    inc  si                            ; i++
    dec  di                            ; j--
_while:
    cmp  si, di                      ; Compara i cu j
    jle  _do                           ; Dacă i <= j se continuă
                                         ; ciclul _do
    cmp  [bp].left, di                ; Compara left cu j
    jge  not_rec                      ; Salt dacă j <= left
                                         ; j > left: se apelează
                                         ; _qasm pentru prima partitură
;
    qasm(v, left, j)
;
    push  di                          ; j
    push  [bp].left                   ; left
    push  bx                          ; v
    call  _qasm
    add   sp, 6
not_rec:
    cmp  si, [bp].right               ; Compara i cu right
    jge  gata                         ; Salt dacă i >= right
                                         ; i < right: se apelează
                                         ; _qasm pentru a doua partitură
;
    qasm(v, i, right)
;
    push  [bp].right                 ; right
    push  si                          ; i
    push  bx                          ; v
    call  _qasm
    add   sp, 6
gata:
    popa
    pop   bp
    ret
_qasm endp
end

```

```
; Interfața cu limbajul C se realizează prin funcția Quick_a,  
; care are un prototip asemănător cu funcția Quick_C.
```

```
; void Quick_a(void *v, size_t n, size_t size, PFCMP cmp);  
{  
    qasm(v, 0, (int) n-1);  
}
```

7.7.3 Compararea performanțelor

Evaluarea performanțelor unui algoritm de sortare internă constă în estimarea sau măsurarea numărului de operații de comparație între elemente și a numărului de operații de copiere sau interschimbare a două elemente. Aceste evaluări se efectuează în trei situații distincte de tablouri: tablou aleator, tablou deja sortat (cazul cel mai favorabil) și tablou sortat invers (cazul cel mai defavorabil).

În cazul de față au fost considerate două dimensiuni de tablouri, de 1000 și respectiv 10000 de întregi, măsurându-se și timpul de execuție.

Pentru algoritmul de sortare rapidă considerat, s-au testat 3 variante: prima (Quick_C) este implementarea polimorfică prezentată în 7.7.1, a doua (Quick_a) este implementarea în ASM din 7.7.3 iar a treia variantă (qsort) este funcția polimorfică din biblioteca standard C.

Rezultatele sunt date în Tabelele 7.2, 7.3 și 7.4.

	Dimensiune = 1000				Dimensiune = 10000			
Alg.	Tim	Comp	Move	Copy	Tim	Comp	Move	Copy
Quick_C	0.3	12901	2443	608	4.3	187234	31508	6219
Quick_a	0.0	12901	2443	0	0.8	187234	31508	0
Qsort	0.2	13321	-	-	2.9	172655	-	-

Tabelul 7.2 Performanțe în cazul unui tablou aleator

	Dimensiune = 1000				Dimensiune = 10000			
Alg.	Tim	Comp	Move	Copy	Tim	Comp	Move	Copy
Quick_C	0.1	8963	488	488	2.3	121821	4095	4095
Quick_a	0.0	8963	488	0	0.4	121821	4095	0
Qsort	0.1	10819	-	-	2.2	145262	-	-

Tabelul 7.3: Performanțe în cazul unui tablou sortat

Alg.	Dimensiune = 1000				Dimensiune = 10000			
	Tim	Comp	Move	Copy	Tim	Comp	Move	Copy
Quick_C	0.2	8972	988	489	2.4	121832	9094	4095
Quick_a	0.0	8972	988	0	0.4	121832	9094	0
Qsort	0.1	10818	-	-	2.3	145260	-	-

Tabelul 7.4: Performanțe în cazul unui tablou sortat invers

Din tabelele de mai sus rezultă că eficiența tuturor celor trei implementări este ridicată. Atunci cînd timpul de rulare este însă esențial, trebuie folosită implementarea specializată în limbaj de asamblare. Implementarea Quick_a asigură un timp de rulare de **3.5 ori mai scurt** în cazul unui fișier aleator și de **5.5 ori mai scurt** în cazul unui fișier sortat sau sortat invers. Toate implementările au utilizat modelul de memorie small.